Proceedings of the International Geoscience and Remote Sensing Symposium IGARSS'99, 28 June - 02 July1999, Hamburg, Germany, ed. Tammy I. Stein, IEEE Cat. No. 99CH36293, ISBN 0-7803-5207-6, IEEE, Piscataway, NJ, USA, 1999.

Horizontal visibility of Lambertian object submerged in seawater

Vladimir I. Haltrin

Naval Research Laboratory, Ocean Sciences Branch, Code 7331, Stennis Space Center, MS 39529-5004, USA
phone: 228-688-4528, fax: 228-688-5379, e-mail: <haltrin@nrlssc.navy.mil>

Abstract — A simple algorithm to calculate horizontal visibility of submerged Lambertian object is proposed. This algorithm is based on the knowledge of the three major parameters: seawater absorption coefficient, seawater forward scattering coefficient, and the albedo of the object. The algorithm is derived from the self-consistent approach to the radiative transfer theory published earlier by the author. This approach is valid for all possible values of absorption and forward scattering coefficients and may be applied without restrictions to any type of coastal water. A FORTRAN code to calculate horizontal visibility is included.

INTRODUCTION

In many predictive visibility models, such as PVM-CWC developed recently by the SAIC and the Scripps Oceanographic Institution [1, 2], the horizontal visibility of a submerged object is estimated only through the beam attenuation coefficient of modeled water. These models ignore such important factors as the forward scattering coefficient by seawater, and the albedo of the object. These omitted properties play a significant role in determining the horizontal visibility of submerged object. This model includes dependence on backscattering coefficient but ignores effects of blurring of the target image associated with the point spread function [3, 4].

APPROACH

The horizontal contrast between Lambertian object with surface albedo A and surrounding seawater background may be expressed as:

$$k_{LO} = |R_A - R_\infty| / R_\infty \qquad (1)$$

here R_A is a horizontal diffuse reflectance of Lambertian object and the layer of water between this object and a viewer, who is supposed to be on the same level with the object, R_∞ is a horizontal diffuse reflectance of the aqueous background.

The basic formulas for horizontal diffuse reflectances may be found in previous papers by the author [5-8]. Some insights on the connection between backscattering coefficient b_B with scattering b and beam attenuation c coefficients may be found in Refs. [8-11].

$$R_A = \frac{R_\infty + \xi \exp(-v x_T)}{1 + R_0\, \xi \exp(-v x_T)}, \quad \xi = \frac{A - R_\infty}{1 - R_0\, A}, \qquad (2)$$

$$R_\infty = \left(\frac{1-\bar\mu}{1+\bar\mu}\right)^2, \quad R_0 = \left(\frac{2+\bar\mu}{2-\bar\mu}\right) R_\infty, \qquad (3)$$

$$v = a\, \frac{7 + 2\bar\mu^2 - \bar\mu^4}{\bar\mu(3-\bar\mu^2)}, \qquad (4)$$

here x_T is a distance between Lambertian target and viewer, $a = c - b$ is an absorption coefficient and $\bar\mu$ is an average cosine determined by the formula [5]:

$$\bar\mu = \sqrt{\frac{a}{a + 3 b_B + \sqrt{b_B (4a + 9 b_B)}}} \qquad (5)$$

In case when only the values of scattering and absorption coefficients are available, we can use the following expression that connects $\bar\mu$ with a single scattering albedo ω_0 [8]:

$$\bar\mu = \sum_{n=0}^{6} c_n (1-\omega_0)^{\frac{n+1}{2}}, \qquad (6)$$

coefficients c_n are given in Ref. [8] or can be fetched from the function `fmuab` listed in the APPENDIX.

The object is visible when $k_{LO} \ge k_h$, here k_h is a threshold contrast for a human eye [12, 13]. By denoting as d_v a maximum distance from which an object can be detected and solving Eq. (1) in respect to $x_T = d_v$, we have the following equation for d_v:

$$d_v = \frac{\bar\mu(3-\bar\mu^2)}{a(7+2\bar\mu^2-\bar\mu^4)} \times \\ \times \ln\left\{\frac{|A - R_\infty|(1 - R_0 R_\infty) - R_0 R_\infty k_h (A - R_\infty)}{k_h R_\infty (1 - A R_0)}\right\}. \qquad (7)$$

Equation (7) is coded in the APPENDIX of this paper.

CONCLUSION

It is shown that the horizontal visibility of a Lambertian object submerged in seawater may be computed on the basis of self-consistent approach in radiative transfer theory [5]. This theory is valid for any arbitrary values of optical properties $a, b,$ and b_B. It means that the approach of this paper can be used in waters of arbitrary turbidity including very turbid coastal waters and waters of lakes and rivers.

ACKNOWLEDGMENT

The author thanks continuing support at the Naval Research Laboratory through the Littoral Optical Environment 6640-09 program. This article represents NRL contribution NRL/PP/7331–99–0001

Proceedings of the International Geoscience and Remote Sensing Symposium IGARSS'99, 28 June - 02 July1999, Hamburg, Germany, ed. Tammy I. Stein, IEEE Cat. No. 99CH36293, ISBN 0-7803-5207-6, IEEE, Piscataway, NJ, USA, 1999.

APPENDIX: FORTRAN FUNCTION HORVIS TO CALCULATE CONTRAST OF LAMBERTIAN OBJECT SUBMERGED IN WATER

```
c ***************************************
      real*8 function horvis
c ***************************************
c     calculates horizontal visibility
c     of an object with Lambertian
c     albedo At placed in water with
c     absorption coefficient a and
c     backscattering coefficient bb.
c     written by Vladimir I. Haltrin
c       <haltrin@nrlssc.navy.mil>
c   The following code is based on the
c   equations proposed in papers:
c    V. I. Haltrin, (See Refs.[5],[8]).
c   input parameters:
c     k,  Visibility contrast limit for
c         the human eye (about 3%)
c     At, target albedo (0 ≤ At ≤ 1.)
c     a,  absorption coefficient in 1/m
c          (arbitrary value)
c     bb, backscattering coefficient in
c         1/m (arbitrary value)
c     b,  scattering coefficient in 1/m
c          (arbitrary value)
c   output parameter:
c   horvis , horizontal visibility
c              distance in m
      implicit none
      real*8   a,bb,At,k
      real*8   fdvis

      call getinput(a,bb,At,k)

      horvis = fdvis(a,bb,At,k)

      return
      end

c =======================================
      real*8 function fmu(a,bb)
c     based on Eq. (37) in Ref. [5]
      implicit none
      real*8   a,bb,q
        q = a+3*bb+SQRT(bb*(4*a+9*bb))
      fmu = SQRT(a/q)
      return
      end

      real*8 function fRb(mu)
c     based on Eq.(41) in Ref. [5]
      implicit none
      real*8   mu,r
        r = (1.-mu)/(1.+mu)
      fRb = r*r
      return
      end

      real*8 function fR0(Rb,mu)
c     based on Eq.(42) in Ref. [5]
      implicit none
      real*8   Rb,mu
      fR0 = (2.+mu)*Rb/(2.-mu)
      return
      end

      real*8 function frnu(a,mu)
c frnu = 1./nu; Eq.(53) in Ref.[5] for nu
      implicit none
      real*8   a,mu,m2
        m2 = mu*mu
      frnu = mu*(3.-m2)/
     &        (a*(7.+m2*(2.-m2)))
      return
      end

      real*8 function fdvis(a,bb,At,k)
c based on Eq.(62) in Ref.[5] and formula
c     for the contrast: k = (Rt-Rb)/Rb
      implicit none
      real*8   a,bb,Rb,R0,mu,R2,Rk,At
      real*8   k,Q,Rq,rnu,dvis
      real*8   fmu,fRb,fR0,frnu
      mu = fmu(a,bb)
      Rb = fRb(mu)
      R0 = fR0(Rb,mu)
      R2 = R0*Rb
      if (At.ge.Rb) then
        if (Rk.ge.1.) then
          fdvis = 0.
c     fdvis is a horizontal visibility
c                distance in m
        else
          Rk = (1.+k)*R2
          Q = k*(1.-R0*At)/(1.-Rk)
          Rq = Rb*Q
          Q = (At-Rb)/Rq
        if ((At.le.Rq).or.(Q.le.1.)) then
           fdvis = 0.
        else
           rnu   = frnu(a,mu)
            fdvis = rnu*LOG(Q)
        end if
        end if
      else                  ! if (At<Rb)
        Rk = (1.-k)*R2
        Q = k*(1.-R0*At)/(1.-Rk)
        Rq = Rb*Q
        Q = (Rb-At)/Rq
        if (Q.le.1.) then
          fdvis = 0.
        else
          rnu   = frnu(a,mu)
          fdvis = rnu*LOG(Q)
        end if
      end if
      return
      end
```

Proceedings of the International Geoscience and Remote Sensing Symposium IGARSS'99, 28 June - 02 July1999, Hamburg, Germany, ed. Tammy I. Stein, IEEE Cat. No. 99CH36293, ISBN 0-7803-5207-6, IEEE, Piscataway, NJ, USA, 1999.

```
      subroutine getinput(a,bb,At,k)
      implicit none
      real*8   a,bb,At,k,b,fbb
      logical  a_and_b
      open(11, file='horvis.in',
     &         status='old')
         read(11,*) a_and_b
         read(11,'(/)') k
c        Visibility contrast limit
c               for the human eye
         read(11,*) At    ! target albedo
         read(11,*) a     ! absorption
c                      coefficient in 1/m
         read(11,*) bb
c       backscattering coefficient in 1/m
         if (a_and_b) then
            read(11,*) b   ! scattering
c                      coefficient in 1/m
            bb = fbb(a,b)
         end if
      close(11)
      return
      end

      real*8 function fmuab(a,b)
c           based on  Eq.(6)  in Ref. [8]
      implicit   none
      real*8     a,b,q
      q = SQRT(a/(a+b))
      fmuab = q*(2.6178398+q*
     &         (-4.6024180+q*(9.0040600+
     &         q*(-14.59994+q*
     &           (14.83909+q*(-8.117954+
     &            1.8593222*q))))))
      return
      end

      real*8 function fgord(mu)
c  based on reversed Eq.(37)  in Ref. [5]
      implicit   none
      real*8     mu,m2,q
        m2 = mu*mu
        q  = 1.-m2
      fgord = q*q/(1.+m2*(4.-m2))
      return
      end

      real*8 function fbb(a,b)
c  calculates backsc. coeff. via a and bb
      implicit   none
      real*8     a,b,mu,g,fmuab,fgord
      mu = fmuab(a,b)     ! average cosine
      g  = fgord(mu) ! Gordon's parameter
      fbb = a*g/(1.-g)    ! backscattering
c     coefficient; see p.3778 in Ref. [5]
      return
      end
```

```
input file "horvis.in"
.false.  a_and_b (if .true. a and b
               are used to calculate bb)
 0.03    k, human eye visib. contr. limit
 0.5     At, target albedo (0 ≤ At ≤ 1.)
 0.2     a, absorption coefficient in 1/m
 0.05    bb, backscattering coeff. in 1/m
 0.3     b,  scattering coeff. in 1/m
```

REFERENCES

[1]. R. R. Hammond, S. A. Jenkins, J. S. Cleveland et al., *Coastal Water Clarity Modeling Feasibility Study*, Final Report, SAIC Technical Report No. 01-1349-03-4841-000 (CDRL A002), February 1995, 650 pp.

[2] S. A. Jenkins, J. Wasyl, R.R.Hammond, Coastal Water Clarity Model: Adaptation for the Hamlet Cove Data Set, SAIC Technical Report, July 1996, 21 pp.

[3] L. S. Dolin, I. M. Levin, *A Handbook of Underwater Vision Theory*, (Leningrad, Gidrometeoizdat, 1991), in Russian

[4] E. P. Zege, A. P. Ivanov and I. L. Katsev, *Image Transfer through a Scattering Media* (Springer Verlag, Berlin, 1991).

[5] V. I. Haltrin, "Self-consistent approach to the solution of the light transfer problem for irradiances in marine waters with arbitrary turbidity, depth and surface illumination," *Appl. Optics*, **37**, 3773-3784 (1998).

[6] V. I. Haltrin, "Apparent optical properties of the sea illuminated by Sun and sky," *Appl. Optics*, **37**, 8336-8340 (1998).

[7] V. I. Haltrin, "Diffuse reflection coefficient of a stratified sea," *Appl. Optics*, **38**, 932-936 (1999).

[8] V. I. Haltrin, "Retrieval of remote radiance reflection coefficients of coastal waters from the inherent optical properties," in 1997 International Geoscience and Remote Sensing Symposium 3-8 August, 1997, Singapore.

[9] V. I. Haltrin, "Theoretical and empirical phase functions for Monte Carlo calculations of light scattering in seawater," in *Proceedings of the Fourth International Conference Remote Sensing for Marine and Coastal Environments*, **I**, pp. 509-518, (Envir. Res. Inst. of Michigan, Ann Arbor, MI, 1997).

[10] V. I. Haltrin, "An analytic Fournier-Forand scattering phase function as an alternative to the Henyey-Greenstein phase function in hydrologic optics," in *1998 IEEE International Geoscience and Remote Sensing Symposium Proceedings*, **II**, T. I. Stein, ed., (IEEE, Piscataway, NJ, USA, 1998), 910-912.

[11] V. I. Haltrin (*a.k.a.* В. И. Халтурин), "Propagation of Light in the Sea Depth," in *Optical Remote Sensing of the Sea and the Influence of the Atmosphere*, V. A. Urdenko and G. Zimmermann, eds., pp. 20-62 (German Democratic Republic Academy of Sciences Institute for Space Research, Berlin, 1985), in Russian.

[12] H. R. Blackwell, "Contrast Thresholds of the human eye," *J. Opt. Soc. Am.*, 36, pp. 624-643, (1946).

[13] J. Williams, *The Meaningful Use of the Secchi Disc*, Technical Report 45, Ref. 68-15, (Chesapeake Bay Institute, The John Hopkins University November 1968).